

Un voyage en Transylvanie

Correction contrôle continu HAI933I - Théorie des bases de données et de connaissances

Le fichier commun.lp vu en cours

```
% une assertion est soit veridique, soit un mensonge,
% mais pas les deux.
veridique(A) :- assertion(A), not mensonge(A).
mensonge(A) :- assertion(A), not veridique(A).
:- assertion(A), veridique(A), mensonge(A).

% une assertion élémentaire exprime un triplet (S P O)
% pour sujet prédicat objet, comme en RDF
assertion(A) :- elementaire(A, S, P, O).
S1 = S2 :- elementaire(A, S1, P1, O1), elementaire(A, S2, P2, O2).
P1 = P2 :- elementaire(A, S1, P1, O1), elementaire(A, S2, P2, O2).
O1 = O2 :- elementaire(A, S1, P1, O1), elementaire(A, S2, P2, O2).
A = B :- elementaire(A, S, P, O), elementaire(B, S, P, O).

% relations entre les assertions élémentaires et les faits
fait(S, P, O) :- elementaire(A, S, P, O), veridique(A).
:- elementaire(A, S, P, O), mensonge(A), fait(S, P, O).
veridique(A) :- elementaire(A, S, P, O), fait(S, P, O).
mensonge(A) :- elementaire(A, S, P, O), not fait(S, P, O).

% conjonctions
assertion(A) :- conjonction(A).
contientMensonge(A) :- element(A, B), mensonge(B).
mensonge(A) :- conjonction(A), contientMensonge(A).
veridique(A) :- conjonction(A), not contientMensonge(A).
veridique(B) :- conjonction(A), veridique(A), element(A, B).
:- conjonction(A), mensonge(A), not contientMensonge(A).

% disjonctions
assertion(A) :- disjonction(A).
contientVeridique(A) :- element(A, B), veridique(B).
veridique(A) :- disjonction(A), contientVeridique(A).
mensonge(A) :- disjonction(A), not contientVeridique(A).
mensonge(B) :- disjonction(A), mensonge(A), element(A, B).
:- disjonction(A), veridique(A), not contientVeridique(A).

% négations
assertion(A) :- negation(A, B).
negation(B, A) :- negation(A, B).
mensonge(A) :- negation(A, B), veridique(B).
veridique(A) :- negation(A, B), mensonge(B).
```

Votre fichier générique transylvanie.lp

```
#include "commun.lp".

% Un transylvanien est un vampire ou un humain, mais pas les deux.
fait(X, isa, humain):- fait(X, isa, transylvanien), not fait(X, isa, vampire).
fait(X, isa, vampire):- fait(X, isa, transylvanien), not fait(X, isa, humain).
:- fait(X, isa, vampire), fait(X, isa, humain).

% Un transylvanien est fou ou un sain d'esprit, mais pas les deux.
fait(X, isa, sain):- fait(X, isa, transylvanien), not fait(X, isa, fou).
fait(X, isa, fou):- fait(X, isa, transylvanien), not fait(X, isa, sain).
:- fait(X, isa, fou), fait(X, isa, sain).

% Un humain sain est honnête, un humain fou est un menteur, etc...
fait(X, isa, honnête) :- fait(X, isa, humain), fait(X, isa, sain).
fait(X, isa, honnête) :- fait(X, isa, vampire), fait(X, isa, fou).
fait(X, isa, menteur) :- fait(X, isa, humain), fait(X, isa, fou).
fait(X, isa, menteur) :- fait(X, isa, vampire), fait(X, isa, sain).
:- fait(X, isa, honnête), fait(X, isa, menteur).

% Tout ce que dit un honnête est vrai, tout ce que dit un menteur est faux
assertion(A), veridique(A) :- fait(X, dit, A), fait(X, is, honnête).
assertion(A), mensonge(A) :- fait(X, dit, A), fait(X, is, menteur).
% et réciproquement
fait(X, isa, honnête) :- fait(X, dit, A), veridique(A).
fait(X, isa, menteur) :- fait(X, dit, A), mensonge(A).

% Pour pouvoir exprimer que dans 2 personnes il y a un humain et un vampire
% (dans les problèmes A)
:- fait(X, autreEspece, Y), fait(X, isa, humain), fait(Y, isa, humain).
:- fait(X, autreEspece, Y), fait(X, isa, vampire), fait(Y, isa, vampire).

% Regles de mariage en transylvanie.
fait(Y, mariageTransylvanien, X) :- fait(X, mariageTransylvanien, Y).
fait(Y, isa, humain):- fait(X, mariageTransylvanien, Y), fait(X, isa, humain).
fait(Y, isa, vampire):- fait(X, mariageTransylvanien, Y), fait(X, isa, vampire).

% Regles d'affichage
res(X, vampire) :- fait(X, isa, vampire).
res(X, humain) :- fait(X, isa, humain).
res(X, sain) :- fait(X, isa, sain).
res(X, fou) :- fait(X, isa, fou).
```

Le problème A1

```
#include "transylvanie.lp".

fait(lucie, isa, transylvanien).
fait(minna, isa, transylvanien).
```

```

fait(lucie, autreEspece, minna).

% Lucie dit "Nous sommes folles"
fait(lucie, dit, a1).
conjonction(a1).
element(a1, a11).
element(a1, a12).
elementaire(a11, lucie, isa, fou).
elementaire(a12, minna, isa, fou).

% Minna répond "Bien sur que non"
fait(minna, dit, a2).
negation(a2, a1).

#show res/2.

```

Answer: 1 res(lucie,vampire) res(minna,humain) res(lucie,sain) res(minna,sain)

Answer: 2 res(lucie,vampire) res(minna,humain) res(lucie,fou) res(minna,fou)

Exemple de réponse: Lucie est un vampire dans tous les modèles stables, nous pouvons donc être certain que Lucie est le vampire.

Le problème A2

```

#include "transylvanie.lp".

fait(aine, isa, transylvanien).
fait(cadet, isa, transylvanien).
fait(aine, autreEspece, cadet).

% L'ainé dit "je suis humain"
fait(aine, dit, a1).
elementaire(a1, aine, isa, humain).

% Le cadet dit "je suis humain"
fait(cadet, dit, a2).
elementaire(a2, cadet, isa, humain).

% L'ainé dit "mon frere est sain d'esprit"
fait(aine, dit, a3).
elementaire(a3, cadet, isa, sain).

#show res/2.

```

Answer: 1 res(cadet,vampire) res(aine,humain) res(aine,sain) res(cadet,sain)

Le problème A3

```
#include "transylvanie.lp".

fait(michael, isa, transylvanien).
fait(peter, isa, transylvanien).
fait(michael, autreEspece, peter).

% Michael dit "je suis un vampire"
fait(michael, dit, a1).
elementaire(a1, michael, isa, vampire).

% Peter dit "je suis un être humain"
fait(peter, dit, a2).
elementaire(a2, peter, isa, humain).

% Michael dit "nous sommes fous tous les 2 ou sains tous les 2"
fait(michael, dit, a3).
disjonction(a3). element(a3, a31). element(a3, a32).
conjonction(a31). element(a31, a311). element(a31, a312).
elementaire(a311, peter, isa, sain).
elementaire(a312, michael, isa, sain).
conjonction(a32). element(a32, a321). element(a32, a322).
elementaire(a321, peter, isa, fou).
elementaire(a322, michael, isa, fou).

#show res/2.
```

Answer: 1 res(peter,vampire) res(michael,humain) res(peter,sain) res(michael,fou)

Le problème A4

```
#include "transylvanie.lp".

fait(pere, isa, transylvanien).
fait(fils, isa, transylvanien).
fait(pere, autreEspece, fils).

% Le pere dit "il y a au moins un fou parmi nous"
fait(pere, dit, a1).
disjonction(a1). element(a1, a11). element(a1, a12).
elementaire(a11, pere, isa, fou).
elementaire(a12, fils, isa, fou).

% Le fils répond "c'est vrai"
fait(fils, dit, a1).

% Le pere dit "je ne suis pas un vampire"
fait(pere, dit, a2).
negation(a2, a3).
elementaire(a3, pere, isa, vampire).

#show res/2.
```

Answer: 1 res(fils,vampire) res(pere,humain) res(pere,sain) res(fils,fou)

Le problème A5

```
#include "transylvanie.lp".

fait(karl, isa, transylvanien).
fait(martha, isa, transylvanien).
fait(karl, autreEspece, martha).

% l'un des 2 est fou et l'autre est sain d'esprit.
:- fait(karl, isa, fou), fait(martha, isa, fou).
:- fait(karl, isa, sain), fait(martha, isa, sain).

% Karl dit "Ma soeur est un vampire"
fait(karl, dit, a1).
elementaire(a1, martha, isa, vampire).

% Martha dit "Mon frere est fou"
fait(martha, dit, a2).
elementaire(a2, karl, isa, fou).

#show res/2.
```

Answer: 1 res(karl,vampire) res(martha,humain) res(karl,sain) res(martha,fou)

Le problème B1

```
#include "transylvanie.lp".

fait(sylvain, isa, transylvanien).
fait(sylvia, isa, transylvanien).
fait(sylvain, mariageTransylvanien, sylvia).

% Sylvia dit "Mon mari est humain"
fait(sylvia, dit, a1).
elementaire(a1, sylvain, isa, humain).

% Sylvain dit "Ma femme est un vampire"
fait(sylvain, dit, a2).
elementaire(a2, sylvia, isa, vampire).

% Sylvia dit "un de nous est fou et l'autre sain"
fait(sylvia, dit, a3).
disjonction(a3). element(a3, a31). element(a3, a32).
conjonction(a31). element(a31, a311). element(a31, a312).
elementaire(a311, sylvain, isa, sain).
elementaire(a312, sylvia, isa, fou).
```

```
conjonction(a32). element(a32, a321). element(a32, a322).  
elementaire(a321, sylvain, isa, fou).  
elementaire(a322, sylvia, isa, sain).  
  
#show res/2.
```

Answer: 1 res(sylvain, humain) res(sylvia, humain) res(sylvia, sain) res(sylvain, fou)

Le problème B2

```
#include "transylvanie.lp".  
  
fait(georges, isa, transylvanien).  
fait(gloria, isa, transylvanien).  
fait(georges, mariageTransylvanien, gloria).  
  
% Gloria dit "tout ce que dit mon mari est vrai"  
fait(gloria, dit, a1).  
elementaire(a1, georges, isa, honnete).  
  
% Georges dit "ma femme est folle"  
fait(georges, dit, a2).  
elementaire(a2, gloria, isa, fou).  
  
#show res/2.
```

Answer: 1 res(georges, vampire) res(gloria, vampire) res(georges, sain) res(gloria, sain)

Answer: 2 res(georges, vampire) res(gloria, vampire) res(georges, fou) res(gloria, fou)

Le problème B3

```
#include "transylvanie.lp".  
  
fait(boris, isa, transylvanien).  
fait(dorothee, isa, transylvanien).  
fait(boris, mariageTransylvanien, dorothee).  
  
% Boris dit "Nous sommes tous les 2 des vampires"  
fait(boris, dit, a1).  
conjonction(a1). element(a1, a11). element(a1, a12).  
elementaire(a11, boris, isa, vampire).  
elementaire(a12, dorothee, isa, vampire).  
  
% Dorothee répond "c'est vrai"  
fait(dorothee, dit, a1).  
  
% Boris dit "nous sommes dans le meme état mental"  
fait(boris, dit, a2).
```

```

disjonction(a2). element(a2, a21). element(a2, a22).
conjonction(a21). element(a21, a211). element(a21, a212).
elementaire(a211, boris, isa, sain).
elementaire(a212, dorothee, isa, sain).
conjonction(a22). element(a22, a221). element(a22, a222).
elementaire(a221, boris, isa, fou).
elementaire(a222, dorothee, isa, fou).

```

```
#show res/2.
```

Answer: 1 res(boris,vampire) res(dorothee,vampire) res(boris,fou) res(dorothee,fou)

Le problème B4

```

#include "transylvanie.lp".

fait(arthur, isa, transylvanien).
fait(liliane, isa, transylvanien).
fait(arthur, mariageTransylvanien, liliane).

% Arthur dit "nous sommes fous"
fait(arthur, dit, a1).
conjonction(a1). element(a1, a11). element(a1, a12).
elementaire(a11, arthur, isa, fou).
elementaire(a12, liliane, isa, fou).

% Liliane répond "c'est vrai"
fait(liliane, dit, a1).

#show res/2.

```

Answer: 1 res(arthur,vampire) res(liliane,vampire) res(arthur,sain) res(liliane,sain)

Answer: 2 res(arthur,vampire) res(liliane,vampire) res(arthur,fou) res(liliane,fou)

Le problème B5

```

#include "transylvanie.lp".

fait(luigi, isa, transylvanien).
fait(manuela, isa, transylvanien).
fait(luigi, mariageTransylvanien, manuela).

% Luigi dit "un de nous 2 au moins est fou"
fait(luigi, dit, a1).
disjonction(a1). element(a1, a11). element(a1, a12).
elementaire(a11, luigi, isa, fou).
elementaire(a12, manuela, isa, fou).

```

```
% Manuela répond "c'est faux"
fait(liliane, dit, a2).
negation(a2, a1).

#show res/2.
```

Answer: 1 res(luigi,vampire) res(manuela,vampire) res(manuela,sain) res(luigi,fou)

Answer: 2 res(luigi,vampire) res(manuela,vampire) res(luigi,sain) res(manuela,sain)

Answer: 3 res(luigi,vampire) res(manuela,vampire) res(luigi,fou) res(manuela,fou)

Answer: 4 res(luigi,humain) res(manuela,humain) res(luigi,sain) res(manuela,fou)

Attention! Smullyan ne donne que la réponse 4, mais les trois autres sont tout aussi correctes... Pour être certain, je suis allé chercher la version anglaise <http://logic-books.info/sites/default/files/lady-or-the-tiger-and-other-logic-puzzles.pdf>, et dans cette version on peut lire Luigi: At least one of us is insane. Manuella: That is not true! Luigi: We are both human. La dernière phrase est absente de la version française! Si on code la version complète on a:

```
#include "transylvanie.lp".

fait(luigi, isa, transylvanien).
fait(manuela, isa, transylvanien).
fait(luigi, mariageTransylvanien, manuela).

% Luigi dit "un de nous 2 au moins est fou"
fait(luigi, dit, a1).
disjonction(a1). element(a1, a11). element(a1, a12).
elementaire(a11, luigi, isa, fou).
elementaire(a12, manuela, isa, fou).

% Manuela répond "c'est faux"
fait(manuela, dit, a2).
negation(a2, a1).

%Luigi says "We are both human" (added from english version).
fait(luigi, dit, a3).
conjonction(a3). element(a3, a31). element(a3, a32).
elementaire(a31, luigi, isa, humain).
elementaire(a32, manuela, isa, humain).

#show res/2.
```

Answer: 1 res(luigi,humain) res(manuela,humain) res(luigi,sain) res(manuela,fou)

L'honneur de Raymond Smullyan est sauf: il ne s'agissait pas d'une erreur logique de sa part, mais d'un simple oubli dans la traduction française.